

Integrity Check for the Stress Incontinence Surgical Treatment Efficacy Trial (SISTER) Data Files

As a partial check of the integrity of the SISTER datasets archived in the NIDDK data repository, a set of tabulations was performed to verify that published results can be reproduced using the archived datasets. Analyses were performed to duplicate results for the data published by Tennstedt et al [1] in the *International Urogynecology Journal and Pelvic Floor Dysfunction* in May 2007. The results of this integrity check are described below. The full text of the *International Urogynecology Journal and Pelvic Floor Dysfunction* article can be found in Attachment 1, and the SAS code for our tabulations is included in Attachment 2.

The intent of this DSIC is to provide confidence that the data distributed by the NIDDK repository is a true copy of the study data. Our intent is *not* to assess the integrity of the statistical analyses reported by study investigators. As with all statistical analyses of complex datasets, complete replication of a set of statistical results should not be expected on a first (or second) exercise in secondary analysis. This occurs for a number of reasons including differences in the handling of missing data, restrictions on cases included in samples for a particular analysis, software coding used to define complex variables, etc. Experience suggests that most discrepancies can ordinarily be resolved by consultation with the study data coordinating center (DCC), however this process is labor-intensive for both DCC and Repository staff. It is thus not our policy to resolve every discrepancy that is observed in an integrity check. Specifically, we do not attempt to resolve minor or inconsequential discrepancies with published results or discrepancies that involve complex analyses, *unless NIDDK Repository staff suspect that the observed discrepancy suggests that the dataset may have been corrupted in storage, transmission, or processing by repository staff*. We do, however, document in footnotes to the integrity check those instances in which our secondary analyses produced results that were not fully consistent with those reported in the target publication.

Background. The Urinary Incontinence Treatment Network (UITN) was set up to establish a group of collaborating investigators who will conduct long-term studies, including clinical trials, of the most commonly used surgical, pharmacological, and behavioral approaches to the management of urinary incontinence in women diagnosed with stress and mixed incontinence.

The UITN consists of nine clinical centers in the U.S. and the biostatistical center is located at the New England Research Institute in Watertown, Massachusetts.

The network's first clinical trial, Stress Incontinence Surgical Treatment Efficacy Results (SISTER), is a randomized controlled clinical trial comparing two surgical procedures commonly used to treat women with stress urinary incontinence; the procedures are the autologous rectus fascial sling procedure and the Burch urethropexy [2].

The Tennstedt paper aims to identify clinical and demographic factors associated with incontinence-related quality of life (QoL) in a cohort of women with stress urinary incontinence who elected surgical treatment [1].

Demographic and Clinical Characteristics. Table 1 in the publication [1] reports on demographic and clinical characteristics of women enrolled in the SISTER Trial. Table A lists the variables we used in our replication.

Table A: Variables Used to Replicate Table 1

Table Variable	Variables Used in Replication
Age	N/A
Ethnicity	Form 01: ethnicity, race_wh, race_bl, race_ai, race_oth
SES	Form 01: occup_code
Frequent UTIs	Form 02: uti_3
Previous UI treatment	Form 02: ui_treat, ui_surg
Anal incontinence	Form 01: sol_stool_lik, oft_solid_lik1, liq_stool_lik, oft_liq_lik1, gas_lik, oft_gas_lik
Solid stool	Form 01: sol_stool_lik, oft_solid_lik1
Liquid stool	Form 01: liq_stool_lik, oft_liq_lik1
Flatus	Form 01: gas_lik, oft_gas_lik
BMI	Form 04: weight, height
Pelvic organ prolapse stage	Form 04: popq_ba, popq_c, popq_bp
Smoking status	Form 01: curr_smoke, reg_smok_age
Stress UI symptoms (MESA)	Form 01: stress_score
Urge UI symptoms (MESA)	Form 01: urge_score
Severity: no. of accidents/day	Form 06: day1_acc, day2_acc, day3_acc
Symptom bother (UDI)	Form 07: qc1-qc20, qc1a-qc20a
Sexually active	Form 07: qe1
Sexual function (PISQ)	Form 07: qf1, qf2, qf3, qf4, qf5, qf6, qf7, qf8, qf9, qf10, qf11, qf12
Note: 'Age' was collected on the Initiation Form, which is not included in the raw datasets.	

In Table B, we compare the results calculated from the archived datasets to the results published in Table 1, Demographic and clinical characteristics of women enrolled in the SISTER Trial. As Table B shows, the results are similar.

Table B: Comparison of Values Computed in Integrity Check to Reference Article Table 1 Values

Potential Correlates	Tennstedt	Integrity Check	Diff
Sociodemographic			
Age, mean (sd)	51.9 (10.3)	---	---
Ethnicity, (n, pct)			
Hispanic	72, 11.0	72, 11.0	0
Non-Hispanic white	480, 73.4	480, 73.3	0, 0.1
Non-Hispanic black	44, 6.7	44, 6.7	0
Non-Hispanic other	58, 8.9	58, 8.9	0
SES, mean (sd)	56.9 (24.6)	57.2 (24.5)	0.3, 0.1
Health status and history, (n, pct)			
Frequent UTIs	45, 6.9	45, 6.9	0
Previous UI treatment	338, 51.6	338, 51.6	0
Anal incontinence	331, 50.5	331, 50.5	0
Solid stool	38, 5.8	38, 5.8	0
Liquid stool	87, 13.3	87, 13.3	0
Flatus	321, 49.0	321, 49.0	0
BMI, mean (sd)	30.0 (6.1)	30.0 (6.1)	0
Pelvic organ prolapse stage, (n, pct)			
0/I	162, 24.7	162, 24.7	0
II	387, 59.1	387, 59.1	0
III/IV	106, 16.2	106, 16.2	0
Smoking status, (n, pct)			
Never	355, 54.2	355, 54.2	0
Former	207, 31.6	207, 31.6	0
Current	93, 14.2	93, 14.2	0
UI type and severity, mean (sd)			
Stress UI symptoms (MESA)	19.4 (4.6)	19.3 (4.6)	0.1, 0
Urge UI symptoms (MESA)	6.5 (3.9)	6.5 (3.9)	0
Severity: no. of accidents/day	3.2 (3.0)	3.2 (3.0)	0
Symptom bother (UDI)	151.0 (48.6)	151.0 (48.6)	0
Sexual activity			
Sexually active, (n, pct)	450, 69.3	450, 68.7	0, 0.6
Sexual function (PISQ), mean (sd)	31.7 (7.0)	31.6 (7.0)	0.1, 0
Notes:			
(1) SES socioeconomic status, UTI urinary tract infection, UI urinary incontinence, BMI body mass index, MESA Medical Epidemiological and Social Aspects of Aging, UDI Urogenital Distress Inventory, PISQ Prolapse/Urinary Incontinence Sexual Questionnaire			

Norma Pugh
October 22, 2008

References

1. Sharon L. Tennstedt et al, **Quality of life in women with stress urinary incontinence**, International Urogynecology Journal and Pelvic Floor Dysfunction; 2007 May; 18(5):543-9.
2. NIDDK Website: The Urinary Incontinence Treatment Network (UITN) page. [UITN: Stress Incontinence Surgical Treatment Efficacy Trial \(SISTEr\)](#).

ATTACHMENT 1

Full Text of Article

Sharon L. Tennstedt et al, Quality of life in women with stress urinary incontinence, International Urogynecology Journal and Pelvic Floor Dysfunction; 2007 May; 18(5):543-9.

NOTE. Single copies of articles published in scientific journals are included with this documentation. These articles are copyrighted, and the repository has purchased ONE reprint from their publisher to include with this documentation. If additional copies are made of these copyrighted articles, users are advised that payment is due to the copyright holder (typically the publisher of the scientific journal).

ATTACHMENT 2

SAS Code for Tabulations from the SISTEr Datasets in the NIDDK Repository

```

/*****
/*
/* Program: R:\05_Users\Norma\SISTER\TennstedtPaper\table1.sas
/* Author:  Norma Pugh
/* Date:    17 November 2008
/* Purpose: Replicate table 1 results.
/*
/*****
/* DATA SOURCE */
libname sasdb 'R:\05_Users\Norma\SISTER\TennstedtPaper\ExtractedData';
libname fmts  'R:\03_Data_And_Tools\Studies\UITN-SISTER\DCC-Delivery';

/* BASELINE FORMATS */
options fmtsearch=(fmts.blformats);
proc format; value yn 1='1=Yes' 2='2=No'; run;

/*****
/* GET STUDY VARIABLES */
/*****
data table1;
  merge sasdb.f01(keep=aid ethnicity race_wh race_bl race_ai race_oth occup_code
sol_stool_lk
                oft_solid_lk1 liq_stool_lk oft_liq_lk1 gas_lk oft_gas_lk curr_smoke
                reg_smok_age stress_score urge_score)
        sasdb.f02(keep=aid uti_3 ui_treat ui_surg)
        sasdb.f04(keep=aid height weight popq_ba popq_c popq_bp)
        sasdb.f06(keep=aid day1_acc day2_acc day3_acc)
        sasdb.f07(keep=aid qc1--qc20a qe1 qf1--qf12);
  by aid;
  /* Ethnicity */
  hispanic=ethnicity;

  if ethnicity=2 then do;
    if race_wh=1 & race_bl=2 & race_ai=2 & race_oth=2 then race='1_White';
    else if race_wh=2 & race_bl=1 & race_ai=2 & race_oth=2 then race='2_Black';
    else race='3_Other';
  end;
  label hispanic='Hispanic?'
        race='Racial group';

  /* SES */
  if occup_code<0 then occup_code=.;

  /* Health status and history */
  if ui_treat=1 or ui_surg=1 then prev_ui_trt=1; else prev_ui_trt=2;
  label prev_ui_trt='Previous UI treatment';

  /* Anal incontinence */
  if sol_stool_lk=1 & oft_solid_lk1 in(2,3) then ai_solid=1; else ai_solid=2;
  if liq_stool_lk=1 & oft_liq_lk1 in(2,3) then ai_liquid=1; else ai_liquid=2;
  if gas_lk=1 & oft_gas_lk in(2,3,4) then ai_flatus=1; else ai_flatus=2;
  if ai_solid=1 or ai_liquid=1 or ai_flatus=1 then ai=1; else ai=2;
  label ai_solid='Anal incontinence: solid stool'
        ai_liquid='Anal incontinence: liquid stool'
        ai_flatus='Anal incontinence: gas'
        ai='Anal incontinence';

```

```

/* Body mass index */
if weight>0 & height>0 then bmi=(weight/height**2)*703;
label bmi='Body mass index';

/* Pelvic organ prolapse stage */
if (popq_ba<-1 & popq_c<-1 & popq_bp<-1) then stage='0/1';
else if (popq_ba>1 or popq_c>1 or popq_bp>1) then stage='3/4';
else stage='2';
label stage='Pelvic organ prolapse stage';

/* Smoking status */
if curr_smoke=1 then smoking='3_Current';
else if reg_smok_age=-2 /* -2=skip */ then smoking='1_Never';
else smoking='2_Former';
label smoking='Smoking status';

/* Severity: no. of accidents/day */
acc=mean (of day1_acc day2_acc day3_acc);
label acc='Number accidents/day';

/* Symptom bother (UDI) */
array udi_old{20}   qc1-qc20;
array udi_old_a{20} qc1a qc2a qc3a qc4a qc5a qc6a qc7a qc8a qc9a qc10a qc11a qc12a qc13a
qc14a qc15a qc16a qc17a qc18a qc19a qc20a;
array udi_comb{20} udi1-udi20;

do u=1 to 20;
  if udi_old{u}=1 then udi_comb{u}=udi_old_a{u};
  else if udi_old{u}=2 then udi_comb{u}=0;
end;

udi_o = 100*(mean(udi5,udi10,udi11,udi12,udi13,udi14,udi15,udi16,udi17,udi18,udi20))/3;
udi_i = 100*(mean(udi1,udi2,udi3,udi7,udi8,udi9))/3;
udi_s = 100*(mean(udi4,udi6))/3;

udi_tot= udi_o + udi_i + udi_s;
label udi_tot='Symptom bother (UDI)';

/* Sexual function (PISQ) */
/* Re-code items from 1-5 to 0-4 Likert scale */
array pisq_old{12} qf1 qf2 qf3 qf4 qf5 qf6 qf7 qf8 qf9 qf10 qf11 qf12;
array pisq_new{12} new_qf1 new_qf2 new_qf3 new_qf4 new_qf5 new_qf6 new_qf7 new_qf8
new_qf9 new_qf10 new_qf11 new_qf12;

do i=1 to 12;
  pisq_new{i}=pisq_old{i}-1;
end;

/* Reverse-score items #1-4 */
array pisq_vars{4} new_qf1 new_qf2 new_qf3 new_qf4;
array pisq_rev{4} rev_qf1 rev_qf2 rev_qf3 rev_qf4;

do j=1 to 4;
  pisq_rev{j}=4-pisq_vars{j};
end;

```

```

/* Calculate sum if 'sex in past 6 months' AND at least 10 of 12 items answered */
n_pisq = n (of rev_qf1 rev_qf2 rev_qf3 rev_qf4 new_qf5 new_qf6 new_qf7 new_qf8 new_qf9
new_qf10 new_qf11 new_qf12);
s_pisq = sum (of rev_qf1 rev_qf2 rev_qf3 rev_qf4 new_qf5 new_qf6 new_qf7 new_qf8
new_qf9 new_qf10 new_qf11 new_qf12);

if qe1=1 & n_pisq >= 10 then pisq=(12/n_pisq)*s_pisq;
label pisq='Sexual function (PISQ)';
run;

/*****/
/* CHECK FOR DUPLICATES */
/*****/
data check_dups; set table1; keep aid; run;
proc sort data=check_dups nodup; by aid; run;

/*****/
/* GET STATISTICS */
/*****/
proc freq data=table1;
tables hispanic race uti_3 prev_ui_trt ai ai_solid ai_liquid ai_flatus stage smoking qe1
/ list missing;
format hispanic uti_3 prev_ui_trt ai_solid ai_liquid ai_flatus ai qe1 yn.;
run;

proc means data=table1 n mean stddev min max; var occup_code bmi stress_score urge_score
acc udi_tot pisq; run;

```